```
> read "HermiteCT.mm";
> with(HermiteCT);
```

[*HermiteReduction*, *HermiteTelescoping*, *KernelReduction*, *PolynomialReduction*, *ShellReduction*] (1)

Calling sequence:
HermiteReduction(H, y)
Input: H, a hyperexponential function in y;
y, a variable name;
Output: [u, r, T], u and r are rational functions in y and T a hyperexponential function with certificate differential-reduced such that

H = Dy(u*T) + r*T,
where r is the residual form of the shell of H.

Example 1 for HermiteReduction

```
> f1 := 1/(y-1)^2;
```

$$f1 := \frac{1}{(y-1)^2} \tag{2}$$

```
> HermiteReduction(f1, y);
```

$$\left[-\frac{1}{y-1}, 0, 1\right] \tag{3}$$

Example 2 for HermiteReduction

```
> f2 := sqrt(y^2+1)/(y-1)^2;
```

$$f2 := \frac{\sqrt{y^2+1}}{(y-1)^2} \tag{4}$$

```
> HermiteReduction(f2, y);
```

$$\left[-\frac{y+1}{2\,(y-1)}, \frac{y\,(y+1)}{2\,(y-1)\,(y^2+1)}, \sqrt{y^2+1}\right] \tag{5}$$

Calling sequence:
HermiteTelescoping(H, x, y, Dx, 'T')
Input: H, a bivariate hyperexponential function;
x, y, two variable names;
Dx, a operator name.
Output: L, a linear differential operator in Dx over F(x) and the fifth argument 'T' is a bivariate hyperexponential function such that

L(x, Dx)(H) = Dy(T).

T is of the form [r, exp(int(udx + vdy)]
with r a rational function and v differential-reduced w.r.t. y.

Example 1 for HermiteTelescoping

```
> h1 := diff(sqrt(x-2*y)*exp(x^2*y), y);
```

$$h1 := -\frac{e^{x^2 y}}{\sqrt{x-2y}} + \sqrt{x-2y}\, x^2\, e^{x^2 y} \tag{6}$$

```
> ZT1 := DETools[Zeilberger](h1, x, y, Dx);
```

$$ZT1 := \left[1, \sqrt{x-2y}\, e^{x^2 y}\right] \tag{7}$$

```
> HT1 := HermiteTelescoping(h1, x, y, Dx, 'T1');
The estimated order of minimal telescopers is 1
```

$$HT1 := 1 \tag{8}$$

Check whether two methods get the same minimal telescoper (up to a univariate rational function in F(x)).

```
> degree(normal(ZT1[1]/HT1), Dx);
```

$$0 \tag{9}$$

```
> T1;
```

$$\left[x-2y, \frac{e^{x^2 y}}{\sqrt{x-2y}}\right] \tag{10}$$

Example 2 for HermiteTelescoping

```
> h2 := sqrt(x-2*y)*exp(x^2*y);
```

$$h2 := \sqrt{x-2y}\, e^{x^2 y} \tag{11}$$

```
> ZT2 := DETools[Zeilberger](h2, x, y, Dx);
```

$$ZT2 := \left[-3x^3 + 6 + 2\, Dx\, x, -\sqrt{x-2y}\,(3x-4y)\, e^{x^2 y}\right] \tag{12}$$

```
> HT2 := HermiteTelescoping(h2, x, y, Dx, 'T2');
The estimated order of minimal telescopers is 1
 Check order 1
```

$$HT2 := -\frac{3(x^3-2)}{2x} + Dx \tag{13}$$

```
> degree(normal(ZT2[1]/HT2), Dx);
```

$$0 \tag{14}$$

Example 3 for HermiteTelescoping

```
> h3 := (1+x+4*x*y)/(11+x+y^2+x*y);
```

$$h3 := \frac{1+x+4xy}{11+x+y^2+xy} \tag{15}$$

```
> ZT3 := DETools[Zeilberger](h3, x, y, Dx):
```

```
> HT3 := HermiteTelescoping(h3, x, y, Dx, 'T3');
```

The estimated order of minimal telescopers is 2 (16)

Check order 1

Check order 2

$$HT3 := -\frac{2\left(3932+53x^2+292x\right)}{\left(-44-4x+x^2\right)\left(6x+44+88x^2+3x^3\right)} + \frac{2x\left(3932+53x^2+292x\right)Dx}{3x^5+76x^4-478x^3-1936-3852x^2-440x} + Dx^2$$

```
> degree(normal(ZT3[1]/HT3), Dx);
```

$$0 \tag{17}$$

Example 4 for HermiteTelescoping

```
> h4 := 1/sqrt(y*(y-1)*(y-x));
```

$$h4 := \frac{1}{\sqrt{y\,(y-1)\,(y-x)}} \tag{18}$$

```
> ZT4 := DETools[Zeilberger](h4, x, y, Dx);
```

$$ZT4 := \left[1+\left(4x^2-4x\right)Dx^2+(8x-4)\,Dx,\ \frac{2y\,(y-1)}{(-y+x)\sqrt{-y\,(y-1)\,(-y+x)}}\right] \tag{19}$$

```
> HT4 := HermiteTelescoping(h4, x, y, Dx, 'T4');
```

The estimated order of minimal telescopers is 2 (20)

Check order 1

Check order 2

$$HT4 := \frac{1}{4x\,(x-1)} + \frac{(2x-1)\,Dx}{x\,(x-1)} + Dx^2$$

```
> degree(normal(ZT4[1]/HT4), Dx);
```

$$0 \tag{21}$$

Example 5 for HermiteTelescoping

```
> h5 := sqrt(x-2*y)*exp(x^2*y);
```

$$h5 := \sqrt{x-2y}\,\mathrm{e}^{x^2 y} \tag{22}$$

```
> ZT5 := DETools[Zeilberger](h5, x, y, Dx);
```

$$ZT5 := \left[-3x^3+6+2\,Dx\,x,\ -\sqrt{x-2y}\,(3x-4y)\,\mathrm{e}^{x^2 y}\right] \tag{23}$$

```
> HT5 := HermiteTelescoping(h5, x, y, Dx, 'T5');
```

The estimated order of minimal telescopers is 1 (24)

Check order 1

$$HT5 := -\frac{3\left(x^3-2\right)}{2x} + Dx$$

```
> degree(normal(ZT5[1]/HT5), Dx);
```

$$0 \tag{25}$$